
1 Introduction

Scalaris is a scalable, transactional, distributed key-value store based on the peer-to-peer principle.
It can be used to build scalable Web 2.0 services. The concept of Scalaris is quite simple: Its
architecture consists of three layers.
It provides self-management and scalability by replicating services and data among peers. Without
system interruption it scales from a few PCs to thousands of servers. Servers can be added or
removed on the fly without any service downtime.

P2P Layer

Transaction Layer 

… scalability

… availability

… strong consistency, 

atomicity, isolation

Replication Layer

Web 2.0 Application
crash 

recovery

model

layer implements …

Many Standard Internet Nodes for Data Storage
crash stop

model

Scalaris takes care of:

• Fail-over

• Data distribution

• Replication

• Strong consistency

• Transactions

The Scalaris project was initiated by Zuse Institute Berlin and onScale solutions and is partly
funded by the EU projects Selfman and XtreemOS. Additional information (papers, videos) can be
found at http://www.zib.de/CSR/Projects/scalaris and http://www.onscale.de/
scalaris.html.

1.1 Brewer’s CAP Theorem

In distributed computing there exist the so called CAP theorem. It basically says that in distributed
systems there are three desirable properties for such systems but can have only any two of them.

Strict Consistency. Any read operation has to return the result of the latest write operation on
the same data item.

Availability. Items can be read and modified at any time.

Partition Tolerance. The network on which the service is running may split into several partitions
which cannot communicate with each other. Lateron the may rejoin again.

For example, a service is hosted on one machine in Seattle and one machine in Berlin. This
service is partition tolerant if it can tolerate that all Internet connections over the Atlantic
(and Pacific) are interrupted for a few hours and then get repaired afterwards.

The goal of Scalarisis to provide strict consistency and partition tolerance. We are willing to
sacrifice availability to make sure that the stored data is always consistent. I.e. when you are
running Scalariswith a replication degree of 4 and the network splits into two partitions, one
partition with three replicas and one partition with one replica, you will be able to continue to use
the service only in the larger partition. All requests in the smaller partition will time out until the
two networks merge again. Note, most other key-value stores tend to sacrifice consistency.

2 Download and Installation

2.1 Requirements

For building and running Scalaris, some third-party modules are required which are not included
in the Scalaris sources:

• Erlang R12 or newer
• GNU-like Make

Note, the Version 12 of Erlang is required. Scalaris will not work with older versions.
To build the Java API (and the command-line client) the following modules are required addition-
ally:

• Java Development Kit 1.6
• Apache Ant

Before building the Java API, make sure that JAVA_HOME and ANT_HOME are set. JAVA_HOME has
to point to a JDK 1.6 installation, and ANT_HOME has to point to an Ant installation.

2.2 Download

The sources can be obtained from http://code.google.com/p/scalaris. RPMs are avail-
able from http://download.opensuse.org/repositories/home:/tschuett/.

2.2.1 Development Branch

You find the latest development version in the svn repository:

# Non-members may check out a read-only working copy anonymously over HTTP.
svn checkout http://scalaris.googlecode.com/svn/trunk/ scalaris-read-only

2.2.2 Releases

Releases can be found under the ’Download’ tab on the web-page.

2.3 Configuration

Scalaris reads two configuration files from the working directory: bin/scalaris.cfg (manda-
tory) and bin/scalaris.local.cfg (optional). The former defines default settings and is in-
cluded in the release. The latter can be created by the user to alter settings. A sample file is
bin/scalaris.local.cfg.example. To run Scalarisdistributed over several nodes, each node
requires a bin/scalaris.local.cfg:

File scalaris.local.cfg:

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Settings for distributed Erlang
% (see cs_send.erl to switch)

% {boot_host, {boot,’boot@foo.bar.com’}}.
% {log_host, {boot_logger, ’boot@foo.bar.com’}}.

%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%
% Settings for TCP mode.
% (see cs_send.erl to switch)

% Insert the appropriate IP-addresses for your setup
% as comma separated integers:
% IP Address, Port, and label of the boot server
{boot_host, {{127,0,0,1},14195,boot}}.

% IP Address, Port, and label of the log server
{log_host, {{127,0,0,1},14195,boot_logger}}.

Scalarisdistinguishes currently two different kinds of nodes: (a) the boot-server and (b) regular
nodes. For the moment, we limit the number of boot-servers to exactly one. The remaining nodes
are regular nodes. The boot-server is contacted to join the system. On all servers, the boot_host
option defines the server where the boot server is running. In the example, it is an IP address plus
a TCP port.

2.4 Build

2.4.1 Linux

Scalaris uses autoconf for configuring the build environment and GNU Make for building the code.

%> ./configure
%> make
%> make docs

For more details read README in the main Scalaris checkout directory.

2.4.2 Windows

We are currently not supporting Scalaris on Windows. However, we have two small bat files for
building and running a boot server. It seems to work but we make no guarantees.
For the most recent description please see the FAQ at http://code.google.com/p/scalaris/
wiki/FAQ.

2.4.3 Java-API

The following commands will build the Java API for Scalaris:

%> make java

This will build scalaris.jar, which is the library for accessing the overlay network. Optionally, the
documentation can be build:

%> cd java-api
%> ant doc

2.5 Running Scalaris

As mentioned above, in Scalaris there are two kinds of nodes:

• boot servers
• regular nodes

In every Scalaris, at least one boot server is required. It will maintain a list of nodes taken part
in the system and allows other nodes to join the ring. For redundancy, it is also possible to have
several boot servers. In the future, we want to eliminate this distinction, so any node is also a
boot-server.

2.5.1 Running on a local machine

Open at least two shells. In the first, go into the bin directory:

%> cd bin
%> ./boot.sh

This will start the boot server. On success http://localhost:8000 should point to the man-
agement interface page of the boot server. The main page will show you the number of nodes
currently in the system. After a couple of seconds a first Scalaris should have started in the boot
server and the number should increase to one. The main page will also allow you to store and
retrieve key-value pairs.
In a second shell, you can now start a second Scalaris node. This will be a ‘regular server’. Go in
the bin directory:

%> cd bin
%> ./cs_local.sh

The second node will read the configuration file and use this information to contact the boot server
and will join the ring. The number of nodes on the web page should have increased to two by now.
Optionally, a third and fourth node can be started on the same machine. In a third shell:

%> cd bin
%> ./cs_local2.sh

In a fourth shell:

%> cd bin
%> ./cs_local3.sh

This will add 3 nodes to the network. The web pages at http://localhost:8000 should show
the additional nodes.

2.5.2 Running distributed

Scalaris can be installed on other machines in the same way as described in Sect. 2.6. In the default
configuration, nodes will look for the boot server on localhost on port 14195. You should create a
scalaris.local.cfg pointing to the node running the boot server.

% Insert the appropriate IP-addresses for your setup
% as comma separated integers:
% IP Address, Port, and label of the boot server
{boot_host, {{127,0,0,1},14195,boot}}.

If you are using the default configuration on the boot server it will listen on port 14195 and you
only have to change the IP address in the configuration file. Otherwise the other nodes will not
find the boot server. On the remote nodes, you only need to call ./cs_local.sh and they will
automatically contact the configured boot server.

2.6 Installation

For simple tests, you do not need to install Scalaris. You can run it directly from the source
directory. Note: make install will install scalaris into /usr/local. But is more convenient to
build RPMs and install those.

svn checkout http://scalaris.googlecode.com/svn/trunk/ scalaris-0.0.1
tar -cvjf scalaris-0.0.1.tar.bz2 scalaris-0.0.1 --exclude-vcs
cp scalaris-0.0.1.tar.bz2 /usr/src/packages/SOURCES/
rpmbuild -ba scalaris-0.0.1/contrib/scalaris.spec

Your source and binary rpm will be generated in /usr/src/packages/SRPMS and RPMS. We
also build rpms using checkouts from svn and provide them using the openSUSE BuildService
at http://download.opensuse.org/repositories/home:/tschuett/. RPM packages are available for

• Fedora 9, 10,

• Mandriva 2008, 2009,

• openSUSE 11.0, 11.1,

• SLE 10, 11,

• CentOS 5 and

• RHEL 5.

Inside those repositories you will also find an erlang rpm - you don’t need this if you already have
a recent enough erlang version!

2.7 Logging

Scalaris uses the log4erl library (see contrib/log4erl for logging status information and error
messages. The log level can be configured in bin/scalaris.cfg. The default value is error;
only errors and severe problems are logged.

%% @doc Loglevel: debug < info < warn < error < fatal < none
{log_level, error}.

In some cases, it might be necessary to get more complete logging information, e.g. for debugging.
In 6.2 on page 23, we are explaining the startup process of Scalarisnodes in more detail, here the
info level provides more detailed information.

%% @doc Loglevel: debug < info < warn < error < fatal < none
{log_level, info}.

3 Using the system

3.1 JSON API

Scalaris supports a JSON API for transactions. To minimize the necessary round trips between
a client and Scalaris, it uses request lists, which contain all requests that can be done in parallel.
The request list is then send over to a Scalaris node with a POST message. The result is an
opaque TransLog and a list containing the results of the requests. To add further requests to the
transaction, the TransLog and another list of requests may be send to Scalaris. This process may
be repeated as necessary. To finish the transaction, the request list can contain a ’commit’ request
as last element, which triggers the validation phase of the transaction processing.
The JSON-API can be accessed via the Scalaris-Web-Server running on port 8000 by default and
the page jsonrpc.yaws (For example at: http://localhost:8000/jsonrpc.yaws). The
following example illustrates the message flow:

Client Scalaris node
Make a transaction, that sets two keys:

{
”method”:” r e q l i s t ”,
” ve r s i on ”:”1.1”,
”params”:
[
[

{ ”wr i t e ”:{”keyA”:”valueA”} },
{ ”wr i t e ”:{”keyB”:”valueB”} },
{ ”commit”:”commit” }

]
],

” id ”:0
}

→

← Scalaris sends results back

{ ” r e s u l t ”:
{ ” r e s u l t s ”:

[
{ ”op”:”commit”,

” va lue ”:”ok”,
”key”:”ok” },

{ ”op”:”wr i t e ”,
” va lue ”:”valueB”,
”key”:”keyB” },

{ ”op”:”wr i t e ”,
” va lue ”:”valueA”,
”key”:”keyA” }

],
” t r an s l og ”:
[...]

},
” id ” : 0

}

In a second transaction: Read the two keys

{
”method”:” r e q l i s t ”,
” ve r s i on ”:”1.1”,
”params”:
[

[
{ ” read”:”keyA” },
{ ” read”:”keyB” }

]
]

” id ”:0
}

→

← Scalaris sends results back

{ ” r e s u l t ”:
{” r e s u l t s ”:
[
{ ”op”:” read”,

” va lue ”:”valueB”,
”key”:”keyB” },

{ ”op”:” read”,
” va lue ”:”valueA”,
”key”:”keyA” }

],
” t r an s l og ”:
[...] // this list is the translog

// for further operations!
// We name it TLOG here.

},
” id ” : 0

}

Calculate something with the read values
and make further requests, here a write and
the commit for the whole transaction. In-
clude also the latest translog we got from
Scalaris (named TLOG here).

{
”method”:” r e q l i s t ”,
” ve r s i on ”:”1.1”,
”params”:
[

TLOG, // translog from prev. result.
[
{ ”wr i t e ”:{”keyA”:”valueA2”} },
{ ”commit”:”commit” }

]
],

” id ” : 0
}

→

← Scalaris sends results back

{ ” r e s u l t ”:
{ ” r e s u l t s ”:

[ { ”op”:”commit”,
” va lue ”:”ok”,
”key”:”ok” },

{ ”op”:”wr i t e ”,
” va lue ”:”valueA2”,
”key”:”keyA” }

],
” t r an s l og ”:
[...]

},
” id ” : 0

}

A sample usage of the JSON API using Ruby can be found in contrib/jsonrpc.rb.
A single request list must not contain a key more than once!
The allowed requests are:

{ ” read”:”any key” }

{ ”wr i t e ”:{”any key”:”any va lue ”} }

{ ”commit”:”commit” }

The possible results are:

{ ”op”:” read”, ”key”:”any key”, ” va lue ”:”any va lue ” }
{ ”op”:” read”, ”key”:”any va lue ”, ” f a i l ”:” reason” } // ’not_found’ or ’timeout’

{ ”op”:”wr i t e ”, ”key”:”any key”, ” va lue ”:”any va lue ” }
{ ”op”:” read”, ”key”:”any key”, ” f a i l ”:” reason” }

{ ”op”:”commit”, ” va lue ”:”ok”, ”key”:”ok” }
{ ”op”:”commit”, ” va lue ”:” f a i l ”, ” f a i l ”:” reason” }

3.1.1 Deleting a key

Outside transactions keys can also be deleted, but it has to be done with care, as explained in
the following thread on the mailing list: http://groups.google.com/group/scalaris/
browse_thread/thread/ff1d9237e218799.

{
”method”:” de l e t e ”,
” ve r s i on ”:”1.1”,
”params”:
[
{ ”key”:”any key” }

],
” id ” : 0

}

Two sample results

{ ” r e s u l t ”:
{ ”ok”:2, // how many replicas were deleted successsfully

” r e s u l t s ”: [ ”ok”, ”ok”, ” l o c k s s e t ”, ”undef” ]
}

}

{ ” r e s u l t ”:
{ ” f a i l u r e ”:” reason” }

}

3.2 Java command line interface

The jar file contains a small command line interface client. For convenience, we provide a wrapper
script called scalaris which setups the Java environment:

%> cd java-api
%> ./scalaris -help
usage: scalaris
-g,--getsubscribers <topic> get subscribers of a topic
-help print this message
-minibench run mini benchmark
-p,--publish <params> publish a new message for a topic: <topic>

<message>
-r,--read <key> read an item
-s,--subscribe <params> subscribe to a topic: <topic> <url>
-u,--unsubscribe <params> unsubscribe from a topic: <topic> <url>
-w,--write <params> write an item: <key> <value>

Read and write can be used to read resp. write from/to the overlay. getsubscribers, publish, and
subscribe are the PubSub functions.

%> ./scalaris -write foo bar
write(foo, bar)
%> ./scalaris -read foo
read(foo) == bar

The scalaris library requires that you are running a ‘regular server’ on the same node. Having a
boot server running on the same node is not sufficient.

3.3 Java API

The scalaris.jar provides the command line client as well as a library for Java programs to
access Scalaris. The library provides two classes:

• Scalaris provides a high-level API similar to the command line client.

• Transaction provides a low-level API to the transaction mechanism.

For details we refer the reader to the Javadoc:

%> cd java-api
%> ant doc
%> firefox doc/index.html

4 Testing the system

4.1 Running the unit tests

There are some unit tests in the test directory. You can call them by running make test in the
main directory. The results are stored in a local index.html file.
The tests are implemented with the common-test package from the Erlang system. For running
the tests we rely on run_test, which is part of the common-test package, but is not installed by
default. configure will check whether run_test is available. If it is not installed, it will show a
warning and a short description of how to install the missing file.
Note: for the unit tests, we are setting up and shutting down several overlay networks. During
the shut down phase, the runtime environment will print extensive error messages. These error
messages do not indicate that tests failed! Running the complete test suite takes about 5 minutes.
Only when the complete suite finished, it will present statistics on failed and successful tests.

5 Troubleshooting

5.1 Network

Scalarisuses a couple of TCP ports for communication. It does not use UDP at the moment.
8000 HTTP Server on the boot node
8001 HTTP Server on the other nodes
14195 Port for inter-node communication (boot server)
14196 Port for inter-node communication (other nodes)

Please make sure that at least 14195 and 14196 are not blocked by firewalls.

6 How a node joins the system

6.1 General Erlang server loop

Servers in Erlang often use the following structure to maintain a state while processing received
messages:

receive
Message ->
State1 = f(State),
loop(State1)

end.

The server runs an endless loop, that waits for a message, processes it and calls itself using tail-
recursion in each branch. The loop works on a State, which can be modified when a message is
handled.

6.2 Starting additional local nodes after boot

After booting a new Scalaris-System as described in Section 2.5.1 on page 11, ten additional local
nodes can be started by typing admin:add_nodes(10) in the Erlang-Shell that the boot process
opened 1.

scalaris/bin> ./boot.sh
[...]
=INFO REPORT==== 12-May-2009::16:24:18 ===
Yaws: Listening to 0.0.0.0:8000 for servers
- http://localhost:8000 under ../docroot
[info] [ CC ] this() == {{127,0,0,1},14195}
[info] [ DNC <0.96.0> ] starting DeadNodeCache
[info] [ DNC <0.96.0> ] starting Dead Node Cache
[info] [ RM <0.97.0> ] starting ring maintainer

[info] [ RT <0.99.0> ] starting routingtable
[info] [ Node <0.101.0> ] joining 315238232250031455306327244779560426902
[info] [ Node <0.101.0> ] join as first 315238232250031455306327244779560426902
[info] [ FD <0.74.0> ] starting pinger for {{127,0,0,1},14195,<0.101.0>}
[info] [ Node <0.101.0> ] joined
[info] [ CY ] Cyclon spawn: {{127,0,0,1},14195,<0.102.0>}
(boot@csr-pc9)1> admin:add_nodes(10)

In the following we will trace, what this function does to join additional nodes to the system.
The function admin:add_nodes(int) is defined as follows.

File admin.erl:

38 %%--------------------------------------------------------------------
39 %% Function: add_nodes(int()) -> ok
40 %% Description: add new Scalaris nodes
41 %%--------------------------------------------------------------------
42 % @doc add new Scalaris nodes on the local node
43 % @spec add_nodes(int()) -> ok
44
45 add_nodes(Count) ->

1Increase the log level to info to get the detailed startup logs. See Sect. 2.7 on page 12

46 add_nodes(Count, 0).
47
48 % @spec add_nodes(int(), int()) -> ok
49 add_nodes(Count, Delay) ->
50 add_nodes_loop(Count, Delay).
51
52 add_nodes_loop(0, _) ->
53 ok;
54 add_nodes_loop(Count, Delay) ->
55 supervisor:start_child(main_sup, {randoms:getRandomId(),
56 {cs_sup_or, start_link, []},
57 permanent,
58 brutal_kill,
59 worker,
60 []}),
61 %timer:sleep(Delay),
62 add_nodes_loop(Count - 1, Delay).

It calls add_nodes_loop(Count, Delay) with a delay of 0. This function starts a new child
for the main supervisor main_sup. As defined by the parameters, to actually perform the start,
the function cs_sup_or:start_link is called by the Erlang supervisor mechanism. For more
details on the OTP supervisor mechanism see Chapter 18 of the Erlang book [1] or the online
documentation at http://www.erlang.org/doc/man/supervisor.html.

6.2.1 Supervisor-tree of a Scalaris node

When starting a new node in the system, the following supervisor tree is created:

One-for-one

supervision

Failure Detector

Configuration

KeyHolder

Statistics Collector

All-for-one

supervision
Chord# Node

Database

Routing Table

Load Balancer

Mod. Paxos

PaxosTransaction

Managers

6.2.2 Starting the or-supervisor and general processes of a node

Starting supervisors is a two step process: the supervisor mechanism first calls the init() func-
tion of the defined module (cs_sup_or:init() in this case) and then calls the start function
(start_link here.
So, lets have a look at cs_sup_or:init, the ’Scalaris or supervisor’.

File cs_sup_or.erl:

61 init([Options]) ->

62 InstanceId = string:concat(” cs node ”, randoms:getRandomId()),
63 boot_server:connect(),
64 KeyHolder =
65 {cs_keyholder,
66 {cs_keyholder, start_link, [InstanceId]},
67 permanent,
68 brutal_kill,
69 worker,
70 []},
71 _RSE =
72 {rse_chord,
73 {rse_chord, start_link, [InstanceId]},
74 permanent,
75 brutal_kill,
76 worker,
77 []},
78 Supervisor_AND =
79 {cs_supervisor_and,
80 {cs_sup_and, start_link, [InstanceId, Options]},
81 permanent,
82 brutal_kill,
83 supervisor,
84 []},
85 RingMaintenance =
86 {?RM,
87
88 {util, parameterized_start_link, [?RM:new(config:read(ringmaintenance_trigger)),
89 [InstanceId]]},
90 permanent,
91 brutal_kill,
92 worker,
93 []},
94 RoutingTable =
95 {routingtable,
96 {util, parameterized_start_link, [rt_loop:new(config:read(routingtable_trigger)),
97 [InstanceId]]},
98 permanent,
99 brutal_kill,

100 worker,
101 []},
102 DeadNodeCache =
103 {deadnodecache,
104 {util, parameterized_start_link, [dn_cache:new(config:read(dn_cache_trigger)),
105 [InstanceId]]},
106 permanent,
107 brutal_kill,
108 worker,
109 []},
110 Vivaldi =
111 {vivaldi,
112 {util, parameterized_start_link, [vivaldi:new(config:read(vivaldi_trigger)),
113 [InstanceId]]},
114 permanent,
115 brutal_kill,
116 worker,
117 []},
118 CS_Reregister =
119 {cs_reregister,
120 {util, parameterized_start_link,[cs_reregister:new(config:read(cs_reregister_trigger)),[InstanceId]]},
121 permanent,
122 brutal_kill,
123 worker,
124 []},
125 DC_Clustering =
126 {dc_clustering,
127 {dc_clustering, start_link, [InstanceId]},
128 permanent,
129 brutal_kill,
130 worker,
131 []},
132 Cyclon =

133 {cyclon,
134 {util, parameterized_start_link, [cyclon:new(config:read(cyclon_trigger)),
135 [InstanceId]]},
136 permanent,
137 brutal_kill,
138 worker,
139 []},
140 Self_Man =
141 {self_man,
142 {self_man, start_link, [InstanceId]},
143 permanent,
144 brutal_kill,
145 worker,
146 []},
147 {ok, {{one_for_one, 10, 1},
148 [
149 Self_Man,
150 CS_Reregister,
151 KeyHolder,
152 RoutingTable,
153 Supervisor_AND,
154 Cyclon,
155 DeadNodeCache,
156 RingMaintenance,
157 Vivaldi
158
159 %,DC_Clustering
160
161 %_RSE
162 ]}}.

The return value of the init() function specifies the child processes of the supervisor and how
to start them. Here, we define a list of processes to be observed by a one_for_one supervi-
sor. The processes are: KeyHolder, DeadNodeCache, RingMaintenance, RoutingTable, and a
Supervisor_AND process.
The term {one_for_one, 10, 1} specifies that the supervisor should try 10 times to restart each
process before giving up. one_for_one supervision means, that if a single process stops, only that
process is restarted. The other processes run independently.
The cs_sup_or:init() is finished and the supervisor module, starts all the defined processes by
calling the functions that were defined in the list of the cs_sup_or:init().
For a join of a new node, we are only interested in the starting of the Supervisor_AND process
here. At that point in time, all other defined processes are already started and running.

6.2.3 Starting the and-supervisor with a peer and its local database

Again, the OTP will first call the init() function of the corresponding module:

File cs_sup_and.erl:

58 init([InstanceId, Options]) ->
59 Node =
60 {cs_node,
61 {cs_node, start_link, [InstanceId, Options]},
62 permanent,
63 brutal_kill,
64 worker,
65 []},
66 DB =
67 {?DB,
68 {?DB, start_link, [InstanceId]},
69 permanent,
70 brutal_kill,
71 worker,
72 []},
73

74 {ok, {{one_for_all, 10, 1},
75 [
76 DB,
77 Node
78 ]}}.

It defines three processes, that have to be observed using an one_for_all-supervisor, which means,
that if one fails, all have to be restarted. Passed to the init function is the InstanceId, a random
number to make nodes unique. It was calculated a bit earlier in the code. Exercise: Try to find
where.
As you can see from the list, the DB is started before the Node. This is intended and important,
because cs_node uses the database, but not vice versa. The supervisor first completely initializes
the DB process and afterwards calls cs_node:start_link. We only go into details here, for the
latter.

File cs_node.erl:

480 %% @doc spawns a scalaris node, called by the scalaris supervisor process
481 %% @spec start_link(term()) -> {ok, pid()}
482 start_link(InstanceId) ->
483 start_link(InstanceId, []).
484
485 start_link(InstanceId, Options) ->
486 gen_component:start_link(?MODULE, [InstanceId, Options], [{register, InstanceId, cs_node}]).

cs_node implements the gen_component behaviour. This component was developed by us to
enable us to write code which is similar in syntax and semantics to the examples in [2]. Similar to
the supervisor behaviour, the component has to provide an init function, but here it is used to
initialize the state of the component. This function is described in the next section.
Note: ?MODULE is a predefined Erlang macro, which expands to the module name, the code belongs
to (here: cs_node).

6.2.4 Initializing a cs_node-process

File cs_node.erl:

471 %% @doc joins this node in the ring and calls the main loop
472 -spec(init/1 :: ([any()]) -> cs_state:state()).
473 init([_InstanceId, _Options]) ->
474 boot_server:be_the_first(),
475 {join_state1}.

The gen_component behaviour registers the cs_node in the process dictionary. Formerly, the
process had to do this himself, but we moved this code into the behaviour. If the cs_node is the
first node, he will start immediately. Otherwise, the process sleeps for a random amount of time.
If you would start 1000 processes with admin:add_nodes(1000), the boot-server would receive
many join requests at the same time, which is not intended. It will also make the ring stabilization
process more complicated. Adding 100s of nodes within a short period of time induces more churn
into the system, than the ring maintenance can handle.
Then, the node retrieves its Id from the keyholder: Id = cs_keyholder:get_key(). In the first
call, a random identifier is returned, otherwise the latest set value. If the cs_node-process failed
and is restarted by its supervisor, this call to the keyholder ensures, that the node still keeps its Id,
assuming that the keyholder process is not failing. This is important for the load-balancing and
for consistent responsibility of nodes to ensure consistent lookup in the structured overlay. Note:
the name Key-holder actually is an id-holder.
If a node changes its position in the ring for load-balancing, the key-holder will be informed and
the cs_node finishes itself. This triggers a restart of the corresponding database process via the

and-supervisor. When the supervisor restarts both processes, they will retrieve the new position
in the ring from the key-holder and join the ring there.
The supervisor was configured to restart a node at most 10 times. Does that
mean, that a node can only change its position in the ring 10 times (caused
by load-balancing)?

6.2.5 Actually joining the ring

After retrieving its identifier, the node starts the join process (cs_join:join).

File cs_join.erl:

The boot-server is contacted to retrieve the known number of nodes in the ring. If the ring is
empty, join_first is called. Otherwise, join_ring is called.
If the ring is empty, the joining node is the only node in the ring and will be responsible for the
whole key space. join_first just creates a new state for a Scalaris node consisting of an empty
routing table, a successorlist containing itself, itself as its predecessor, a reference to itself, its
responsibility area from Id to Id (the full ring), and a load balancing schema.

File cs_join.erl:

The macro ?RT maps to the configured routing algorithm and ?RM to the configured ring mainte-
nance algorithm. It is defined in include/scalaris.hrl. For further details on the routing see
Chapter 7 on page 31.
The state is defined in

File cs_state.erl:

67 new(RT, Successor, Predecessor, Me, MyRange, LB, DB) ->
68 #state{
69 routingtable = RT,
70 successor = Successor,
71 predecessor = Predecessor,
72 me = Me,
73 my_range = MyRange,
74 lb=LB,
75 join_time=now(),
76 deadnodes = gb_sets:new(),
77 trans_log = #translog{
78 tid_tm_mapping = dict:new(),
79 decided = gb_trees:empty(),
80 undecided = gb_trees:empty()
81 },
82 db = DB
83 }.

If a node joins an existing ring, reliable_get_node is called for the own Id in cs_join:join().
This lookup delivers the node who is currently responsible for the new node’s identifier – the
successor for the joining node. If this lookup fails for some reason, it is tried again, by recursivly
calling the join().
What, if the Id is exactly the same as that of the existing node? This could
lead to lookup and responsibility inconsistency? Can this be triggered by
the load-balancing? This is a bug, that should be fixed!!!
Then, cs_join:join_ring is called:

File cs_join.erl:

First the node is initialized. Then it sends a join message to the successor including a reference
to itself and the chosen Id.
The message is received by the old node in cs_node.erl. There exists a {join, X} handler.

File cs_node.erl:

462 on({join, Source_PID, Id, UniqueId}, State) ->
463 cs_join:join_request(State, Source_PID, Id, UniqueId);
464
465
466
467 on(_, _State) ->
468 unknown_event.
469
470 %% userdevguide-begin cs_node:start
471 %% @doc joins this node in the ring and calls the main loop
472 -spec(init/1 :: ([any()]) -> cs_state:state()).
473 init([_InstanceId, _Options]) ->
474 boot_server:be_the_first(),
475 {join_state1}.
476
477 %% userdevguide-end cs_node:start
478
479 %% userdevguide-begin cs_node:start_link
480 %% @doc spawns a scalaris node, called by the scalaris supervisor process
481 %% @spec start_link(term()) -> {ok, pid()}
482 start_link(InstanceId) ->
483 start_link(InstanceId, []).
484
485 start_link(InstanceId, Options) ->
486 gen_component:start_link(?MODULE, [InstanceId, Options], [{register, InstanceId, cs_node}]).
487 %% userdevguide-end cs_node:start_link
488
489 get_local_cyclon_pid() ->
490 InstanceId = erlang:get(instance_id),
491 if
492 InstanceId == undefined ->
493 log:log(error,”[ Node ] ˜p”, [util:get_stacktrace()]);
494 true ->
495 ok
496 end,
497 process_dictionary:lookup_process(InstanceId, cyclon).
498
499 get_local_cs_reregister_pid() ->
500 InstanceId = erlang:get(instance_id),
501 if
502 InstanceId == undefined ->
503 log:log(error,”[ Node ] ˜p”, [util:get_stacktrace()]);
504 true ->
505 ok
506 end,
507 process_dictionary:lookup_process(InstanceId, cs_reregister).

This triggers a call to join_request on the old node.

File cs_join.erl:

39 join_request(State, Source_PID, Id, UniqueId) ->
40 Pred = node:new(Source_PID, Id, UniqueId),
41 {DB, HisData} = ?DB:split_data(cs_state:get_db(State), cs_state:id(State), Id),
42 cs_send:send(Source_PID, {join_response, cs_state:pred(State), HisData}),
43 ring_maintenance:update_pred(Pred),
44 cs_state:set_db(State, DB).

The cs_node notifies the ring maintenance, that he has a new predecessor. Then he removes the
key-value pairs from his database which are now in the responsibility of the joining node. Then it
sends a join_response to the new node with its former predecessor, the data, it has to host, and
its successorlist.

Back on the joining node: it waits for the join_response message in cs_join:join_ring().
The next steps after the message was received from the old node are to initialize the maintenance
components for the ring and routing table, the database and the state of the cs_node.

6.2.6 Beginning to serve requests

cs_join:join() was called from cs_node:start(), which now continues

File cs_node.erl:

471 %% @doc joins this node in the ring and calls the main loop
472 -spec(init/1 :: ([any()]) -> cs_state:state()).
473 init([_InstanceId, _Options]) ->
474 boot_server:be_the_first(),
475 {join_state1}.

The cs_replica_stabilization:recreate_replicas() function is called, which is not yet
implemented. It would recreated necessary replicas that were lost due to load-balancing and node
failures.
Finally, the loop for request handling is started.

7 Routing and routing tables in the Overlay

Each node of the ring can perform searches in the overlay.
A search is done by a lookup in the overlay, but there are several other demands for communication
between peers, so Scalaris provides a general interface to route a message to another peer, that is
currently responsible for a given key.

File cs_lookup.erl:

[...]
unreliable_lookup(Key, Msg) ->

get_pid(cs_node) ! {lookup_aux, Key, Msg}.

unreliable_get_key(Key) ->
unreliable_lookup(Key, {get_key, cs_send:this(), Key}).

[...]

The message Msg could be a get which retrieves content from the responsible node or a get_node
message, which returns a pointer to the node.
All currently supported messages are listed in the file cs_node.erl.
The message routing is implemented in lookup.erl

File lookup.erl:

[...]
lookup_fin(Msg) ->

self() ! Msg.

lookup_aux(State, Key, Msg) ->
Terminate = util:is_between(cs_state:id(State), Key, cs_state:succ_id(State)),
P = ?RT:next_hop(State, Key),
?LOG(”[ ˜w | I | Node | ˜w ] lookup aux ˜w ˜w ˜s˜n”,
[calendar:universal_time(), self(), Terminate, P, Key]),

if
Terminate ->

cs_send:send(P, {lookup_fin, Msg});
true ->

cs_send:send(P, {lookup_aux, Key, Msg})
end.

[...]

Each node is responsible for a certain key interval. The function util:is_between is used to
decide, whether the key is between the current node and its successor. If that is the case, final
step is done using lookup_fin(), which delivers the message to the local node. Otherwise, the
message is forwarded to the next nearest known peer (listed in the routing table) determined by
?RT:next_hop.
routingtable.erl is a generic interface for routing tables. It can be compared to interfaces in
Java. In Erlang interfaces can be defined using a so called ‘behaviour’. The files rt_simple and
rt_chord implement the behaviour ‘routingtable’.
The macro ?RT is used to select the current implementation of routing tables. It is defined in
scalaris.hrl.

File scalaris.hrl:

26 %%This file determines which kind of routingtable is used. Uncomment the
27 %%one that is desired.

28
29 %%Standard Chord routingtable
30 -define(RT, rt_chord).
31
32 %%Simple routingtable
33 %-define(RT, rt_simple).

The functions, that have to be implemented for a routing mechanism are defined in the following
file:

File routingtable.erl:

42 behaviour_info(callbacks) ->
43 [
44 % create a default routing table
45 {empty, 1},
46 % mapping: key space -> identifier space
47 {hash_key, 1}, {getRandomNodeId, 0},
48 % routing
49 {next_hop, 2},
50 % trigger for new stabilization round
51 {init_stabilize, 3},
52 % dead nodes filtering
53 {filterDeadNode, 2},
54 % statistics
55 {to_pid_list, 1}, {get_size, 1},
56 % for symmetric replication
57 {get_keys_for_replicas, 1},
58 % for debugging
59 {dump, 1},
60 % for bulkowner
61 {to_dict, 1},
62 % convert from internal representation to version for cs_node
63 {export_rt_to_cs_node, 4},
64 % update pred/succ in routing stored in cs_node
65 {update_pred_succ_in_cs_node, 3}
66 ];

empty/1 gets a successor passed and generates an empty routing table. The data structure of the
routing table is undefined. It can be a list, a tree, a matrix . . .

hash_key/1 gets a key and maps it into the overlay’s identifier space.
getRandomNodeId/0 returns a random node id from the overlay’s identifier space. This is used

for example when a new node joins the system.
next_hop/2 gets a routing table and a key and returns the node, that should be contacted next

(is nearest to the id).
init_stabilize/3 is called periodically to rebuild the routing table. The parameters are the

identifier of the node, the successor and the old routing table state.
filterDeadNode/2 is called by the failuredetector and tells the routing table about dead nodes

to be eliminated from the routing table. This function cleans the routing table.
to_pid_list/1 get all PIDs of the routing table entries.
get_size/1 get the routing table’s size.
get_keys_for_replicas/1 Returns for a given Key the keys of its replicas. This used for

implementing symmetric replication.
dump/1 dump the state. Not mandatory, may just return ok.
to_dict/1 returns the routing tables entries in an array-like structure. This is used by bulk-

operations to create a broadcast tree.

7.1 Simple routing table

One implementation of a routing table is the rt_simple, which routes via the successor, which is
inefficient, as it needs a linear number of hops to reach its goal. A more robust implementation,
would use a successor list. This implementation is not very efficient on churn.

7.1.1 Data types

First, the data structure of the routing table is defined:

File rt_simple.erl:

39 % @type key(). Identifier.
40 -type(key()::pos_integer()).
41 % @type rt(). Routing Table.
42 -ifdef(types_are_builtin).
43 -type(rt()::{node:node_type(), gb_tree()}).
44 -type(external_rt()::{node:node_type(), gb_tree()}).
45 -else.
46 -type(rt()::{node:node_type(), gb_trees:gb_tree()}).
47 -type(external_rt()::{node:node_type(), gb_trees:gb_tree()}).
48 -endif.

A routing table is a pair of a node (the successor) and an (unused) gb_tree. Keys in the overlay
are identified by integers.

7.1.2 A simple routingtable behaviour

File rt_simple.erl:

52 %% @doc creates an empty routing table.
53 %% per default the empty routing should already include
54 %% the successor
55 -spec(empty/1 :: (node:node_type()) -> rt()).
56 empty(Succ) ->
57 {Succ, gb_trees:empty()}.

The empty routing table consists of the successor and an empty gb_tree.

File rt_simple.erl:

61 %% @doc hashes the key to the identifier space.
62 -spec(hash_key/1 :: (any()) -> key()).
63 hash_key(Key) ->
64 BitString = binary_to_list(crypto:md5(Key)),
65 % binary to integer
66 lists:foldl(fun(El, Total) -> (Total bsl 8) bor El end, 0, BitString).

Keys are hashed using MD5 and have a length of 128 bits.

File rt_simple.erl:

77 %% @doc returns the next hop to contact for a lookup
78 -spec(next_hop/2 :: (cs_state:state(), key()) -> pid()).
79 next_hop(State, _Key) ->
80 cs_state:succ_pid(State).

Next hop is always the successor.

File rt_simple.erl:

84 %% @doc triggered by a new stabilization round
85 -spec(init_stabilize/3 :: (key(), node:node_type(), rt()) -> rt()).

86 init_stabilize(_Id, Succ, _RT) ->
87 % renew routing table
88 empty(Succ).

init_stabilize/3 resets its routing table with the current successor.

File rt_simple.erl:

92 %% @doc removes dead nodes from the routing table
93 -spec(filterDeadNode/2 :: (rt(), cs_send:mypid()) -> rt()).
94 filterDeadNode(RT, _DeadPid) ->
95 RT.

filterDeadNodes/2 does nothing, as only the successor is listed in the routing table and that is
reset periodically in init_stabilize/3.

File rt_simple.erl:

99 %% @doc returns the pids of the routing table entries .
100 -spec(to_pid_list/1 :: (rt()) -> [cs_send:mypid()]).
101 to_pid_list({Succ, _RoutingTable} = _RT) ->
102 [node:pidX(Succ)].

to_pid_list/1 returns the pids of the routing tables, as defined in node.erl.

File rt_simple.erl:

111 normalize(Key) ->
112 Key band 16#FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF.
113
114 n() ->
115 16#100000000000000000000000000000000.
116
117 %% @doc returns the replicas of the given key
118 -spec(get_keys_for_replicas/1 :: (key() | string()) -> [key()]).
119 get_keys_for_replicas(Key) when is_integer(Key) ->
120 [Key,
121 normalize(Key + 16#40000000000000000000000000000000),
122 normalize(Key + 16#80000000000000000000000000000000),
123 normalize(Key + 16#C0000000000000000000000000000000)
124 ];
125 get_keys_for_replicas(Key) when is_list(Key) ->
126 get_keys_for_replicas(hash_key(Key)).

The get_keys_for_replicas/1 implements symmetric replication, here. The call to normalize
implements the modulo by throwing high bits away.

File rt_simple.erl:

131 %% @doc
132 -spec(dump/1 :: (rt()) -> ok).
133 dump(_State) ->
134 ok.

dump/1 is not implemented.

7.2 Chord routing table

The file rt_chord.erl implements Chord’s routing.

7.2.1 Data types

File rt_chord.erl:

40 -type(key()::pos_integer()).
41 -ifdef(types_are_builtin).
42 -type(rt()::gb_tree()).
43 -type(external_rt()::gb_tree()).
44 -type(dict_type() :: dict()).
45 -else.
46 -type(rt()::gb_trees:gb_tree()).
47 -type(external_rt()::gb_trees:gb_tree()).
48 -type(dict_type() :: dict:dictionary()).
49 -endif.
50 -type(index() :: {pos_integer(), pos_integer()}).
51
52 -type(state()::rt()).

The routing table is a gb_tree. Identifiers in the ring are integers. Note, that in Erlang integer
can be of arbitrary precision. For Chord, the identifiers are in [0, 2128), i.e. 128-bit strings.

7.2.2 The routingtable behaviour for Chord

File rt_chord.erl:

60 %% @doc creates an empty routing table.
61 -spec(empty/1 :: (node:node_type()) -> rt()).
62 empty(_Succ) ->
63 gb_trees:empty().

empty/1 returns an empty gb_tree.
hash_key(Key) and getRandomNodeId call their counterparts from rt_simple.erl

File rt_chord.erl:

195 %% @doc returns the next hop to contact for a lookup
196 %% Note, that this code will be called from the cs_node process and
197 %% it will have an external_rt!
198 -spec(next_hop/2 :: (cs_state:state(), key()) -> cs_send:mypid()).
199 next_hop(State, Id) ->
200 case util:is_between(cs_state:id(State), Id, cs_state:succ_id(State)) of
201 %succ is responsible for the key
202 true ->
203 cs_state:succ_pid(State);
204 % check routing table
205 false ->
206 case util:gb_trees_largest_smaller_than(Id, cs_state:rt(State)) of
207 nil ->
208 cs_state:succ_pid(State);
209 {value, _Key, Value} ->
210 Value
211 end
212 end.

next_hop traverses the routing table beginning with the longest finger (2127) by calling the helper
function next_hop/5.

File rt_chord.erl:

If the entry exists, it is retrieved from the gb_tree. If the id of the routing table entry is between
ourselves and the searched id, the finger is chosen. If anything fails, Candidate (the successor) is
chosen.
Why could a routing table entry be null? filterDeadNodes changes entries to null.

BUG: Instead of directly returning Candidate one should further traverse the
routing table for shorter appropriate fingers. If doing so, a check whether
Index is zero, would become necessary.
If the finger is to long, recursively try the next shorter finger.

File rt_chord.erl:

82 %% @doc starts the stabilization routine
83 -spec(init_stabilize/3 :: (key(), node:node_type(), rt()) -> rt()).
84 init_stabilize(Id, _Succ, RT) ->
85 % calculate the longest finger
86 Key = calculateKey(Id, first_index()),
87 % trigger a lookup for Key
88 cs_lookup:unreliable_lookup(Key, {rt_get_node, cs_send:this(), first_index()}),
89 RT.

The routing table stabilization is triggered with the index 127 and then runs asynchronously, as we
do not want to block the rt_loop to perform other request while recalculating the routing table.
We have to find the node responsible for the calculated finger and therefore perform a lookup for
the node with a rt_get_node message, including a reference to ourselves as the reply-to address
and the index to be set.
The lookup performs an overlay routing by passing the massage until the responsible node is
found. There, the message is delivered to the cs_node. At the destination the message is handled
in cs_node.erl:

File cs_node.erl:

297 on({rt_get_node, Source_PID, Cookie}, State) ->
298 cs_send:send(Source_PID, {rt_get_node_response, Cookie, cs_state:me(State)}),
299 State;

The remote node just sends the requested information back directly in a rt_get_node_response
message including a reference to itself. When receiving the routing table entry, we call stabilize/5.

File rt_chord.erl:

124 %% @doc updates one entry in the routing table
125 %% and triggers the next update
126 -spec(stabilize/5 :: (key(), node:node_type(), rt(), pos_integer(),
127 node:node_type()) -> rt()).
128 stabilize(Id, Succ, RT, Index, Node) ->
129 case node:is_null(Node) % do not add null nodes
130 orelse (node:id(Succ) == node:id(Node)) % there is nothing shorter than succ
131 orelse (util:is_between(Id, node:id(Node), node:id(Succ))) of % there should not be anything shorter than succ
132 true ->
133 RT;
134 false ->
135 NewRT = gb_trees:enter(Index, Node, RT),
136 Key = calculateKey(Id, next_index(Index)),
137 cs_lookup:unreliable_lookup(Key, {rt_get_node, cs_send:this(),
138 next_index(Index)}),
139 NewRT
140 end.

stabilize/5 assigns the received routing table entry and triggers to fill the next shorter one using
the same mechanisms as described.
When the shortest finger is the successor, then filling the routing table is stopped, as no further
new entries would occur. It is not necessary, that Index reaches 1 to make that happen. If less
than 2128 nodes participate in the system, it may happen earlier.
filterDeadNode removes dead entries from the gb_tree.

File rt_chord.erl:

93 %% @doc remove all entries
94 -spec(filterDeadNode/2 :: (rt(), cs_send:mypid()) -> rt()).
95 filterDeadNode(RT, DeadPid) ->
96 DeadIndices = [Index|| {Index, Node} <- gb_trees:to_list(RT),
97 node:pidX(Node) == DeadPid],
98 lists:foldl(fun (Index, Tree) -> gb_trees:delete(Index, Tree) end,
99 RT, DeadIndices).

8 Directory Structure of the Source Code

The directory tree of Scalaris is structured as follows:

bin contains shell scripts needed to work with Scalaris (e.g. start the boot
services, start a node, . . . )

contrib necessary third party packages (yaws and log4erl)
doc generated erlang documentation

docroot root directory of the bootserver’s webserver
docroot_node root directory of the normal node’s webserver

ebin the compiled Erlang code (beam files)
java-api a java api to Scalaris

log log files
src contains the Scalaris source code
test unit tests for Scalaris

user-dev-guide contains the sources for this document

9 Java API

For the Java API documentation, we refer the reader to Javadoc resp. doxygen. The following
commands create the documentation:

%> cd java-api
%> ant doc
%> doxygen

The Javadoc can be found in java-api/doc/index.html. The doxygen is in
doc-doxygen/html/index.html.
We provide two kinds of APIs:

• high-level access with de.zib.scalaris.Scalaris

• low-level access with de.zib.scalaris.Transaction

The former provides general functions for reading and writing single key-value pairs and an API
for the built-in PubSub-service. The latter allows the user to write custom transactions which can
modify an arbitrary number of key-value pairs within one transaction.

